INTRODUKTION TILL MAPLE

Kurt Hansson

26 november 2003

Innehåll

1 Inledning

Datorer kan snabbt och med stor säkerhet utföra beräkningar enligt bestämda regler (algoritmer). Man har därför länge kunnat använda dem för beräkningar med rationella tal. De har varit svårare att utnyttja för allmännare beräkningar – derivering av funktioner, faktorisering av polynom, partialbråksutveckling, m.m. De största svårigheterna har varit att finna bra representationer för storheterna, att ange bra och snabba algoritmer samt att förenkla resultaten. Det finns nu ett antal program (Maple, Mathematica, Axiom, ...) som kan underlätta arbetsamma beräkningar, men fortfarande är det ofta problem med förenklingar.

2 Datorhantering

2.1 Start av Maple

För användare av datorerna i sjösystemet finns utförliga anvisningar för start och utskrifter på www-sidan:

`http://www.mai.liu.se/TM/kurser/TATM72/maple_at_sea.html`

För att starta Maple vid en sunstation (med solaris) i t.ex. någon av ISY-datorsalarna öppnar man ett terminalfönster och skriver

```
module add maple <RETUR>
xmaple <RETUR>
```

Maple startar efter en stund och man hamnar i det grafiska gränssnittet. Skriver man bara `maple` i sista kommandot får man ett tty-gränssnitt som möjligen kan vara intressant för gamla datorpurister som är rädda för möss. Vill man inte köra defaultversionen kan man istället skriva

```
module add maple/8 <RETUR>
xmaple <RETUR>
```

Då startas version 8 (som förefaller stabilare än version 9).

2.2 Utskrift

Innan man skriver ut sina beräkningar tar man först bort allt som inte skall vara med och skriver sedan in lämpliga kommentarer i text-fält i anslutning till beräkningarna. Därefter väljer man Print i File-menyn. Då öppnas ett nytt fönster och man kan välja om man vill spara dokumentet som en ps-fil för senare utskrift eller om man skall skicka direkt till skrivaren. Se också till att rätt pappersformat (European A4) är valt.

OBS! *Vid tentamen är det mycket viktigt att bladen förses med p-nummer och namn.*

3 Att använda Maple

För en utförligare introduktion till Maple rekommenderas boken [2]. Vi skall mycket kortfattat beskriva hur man kan använda Maple. Hur skriver man då kommandon i Maple?

3.1 Räkneoperationer

Operation	**Skrivs**	**Operation**	**Skrivs**
Addition	`a + b`	Division	`a/b`
Subtraktion	`a - b`	a upphöjt till b	`a^b` eller `a**b`
Multiplikation	`a * b`	Decimalkomma	`.` (d v s decimalpunkt)

3.2 Konstanter och elementära funktioner

Observera att vinkeln, i trigonometriska uttryck, skall anges i radianer.

Funktioner och konstanter	**Skrivs**
Trigonometriska funktioner	`sin(x)`, `cos(x)`, `tan(x)`, `cot(x)`
Arcusfunktioner	`arcsin(x)`, `arccos(x)`, `arctan(x)`, `arccot(x)`
Exponentialfunktionen e^x	`exp(x)`
Naturliga logaritmen $\ln x$	`ln(x)`
Kvadratroten $\sqrt{x}$	`sqrt(x)`
Absolutbeloppet $\|x\|$	`abs(x)`
Binomialkoefficienten $\binom{n}{k}$	`binomial(n,k)`
Hyperboliska funktioner	`sinh(x)`, `cosh(x)`, `tanh(x)`, `coth(x)`
Konstanterna e, π och i	`exp(1),` `Pi` respektive `I`

3.3 Kommandon och resultat

När programmet är redo för ett kommando ges en kil, > , i början av raden. Kommandot skrivs in och avslutas med semikolon och <retur> för att Maple skall utföra kommandot och skriva ut resultatet. Ger du inget semikolon tolkar Maple detta som att kommandot fortsätter på nästa rad. Har du skrivit in ett kommando och tryckt på <retur>-knappen och därefter upptäcker att du glömt semikolon löser du detta problem enkelt, du skriver ett semikolon och <retur>. Nedan följer några exempel. Efter kilen står de kommandon som matats in. Sedan kommer de resultat som Maple producerat. Jag har också skrivit in en del kommentarer till dessa kommandon och resultat.

Multiplikation

```
> 45*39;
```

$$1755$$

Numerisk beräkning blir det när man använder decimaltal

```
> 12.5*1.38/9.62;
```

$$1.793139293$$

För att bilda talet $5a$ måste vi använda `*`

```
> 5*a;
```

$$5\,a$$

Division och multiplikation har högre prioritet än addition och subtraktion

```
> a+b/a-b;
```

$$a+\frac{b}{a}-b$$

Parenteser måste användas om addition och subtraktion ska utföras innan multiplikation respektive division

```
> (a+b)/(a-b);
```

$$\frac{a+b}{a-b}$$

Här blir det fel!

```
> sin(2*cos(x^(3|y)));
```

```
missing operator or ‘;‘
```

$3y$ måste skrivas `3*y` för att Maple ska förstå. Markören hamnar där felet är. Så här skall det alltså se ut

```
> sin(2*cos(x^(3*y)));
```

$$\sin(2\cos(x^{(3\,y)}))$$

Talet π skrivs `Pi` och $\sqrt{81}=9$ vilket Maple ”vet”

```
> Pi*sqrt(81);
```

$$9\,\pi$$

Här blir det fel igen!

```
> cos(Pi/3)
> ;
```

```
Warning, premature end of input
```

Semikolon saknas. Skriv `;` och <`retur`> vid den nya kilprompten för att kommandot skall vara avslutat. Att $\cos(\pi/3)=1/2$ är ”känt” av Maple.

```
> I^2;
```

$$-1$$

Beräknar i^2. Notera att det imaginära talet i skrivs `I`.

```
> exp(ln(3));
```

$$3$$

Beräknar $e^{\ln 3}$.

3.4 Definiera uttryck och funktioner

För att ge ett uttryck ett namn används operationen `:=` (tilldelning) Följande kommando gör att uttrycket $(x-1)x+e^x$ tilldelas namnet f.

```
> f := (x-1)*x+exp(x);
```

$$f := (x-1)\,x+e^x$$

Vill man senare arbeta med detta uttryck räcker det alltså att använda sig av symbolen f. För att ersätta x med t ex 3 i uttrycket skriver man (se listan nedan).

```
> subs(x=3, f);
```

$$6+e^3$$

För att definiera en funktion är det ibland mer praktiskt att använda operationen `:= x ->` Så här definierar man funktionen $f(x)=(x-1)x+e^x$.

```
> f := x -> (x-1)*x+exp(x);
```

$$f := x \to (x-1)\,x+e^x$$

Funktionen anropar man med `f(x);`. Skriver man t ex `f(3);` svarar Maple med funktionens värde för $x=3$.

```
> f(3);
```

$$6+e^3$$

`f(t);` går också bra

```
> f(t);
```

$$(t-1)\,t+e^t$$

Har man redan ett uttryck

```
> f := x*sin(x)+exp(x);
```

$$f := x\sin(x)+e^x$$

och vill göra om det till en funktion g kan man använda funktionen `unapply`

```
> g := unapply(f, x);
```

$$g := x \to x\sin(x)+e^x$$

```
> g(7);
```

$$7\sin(7)+e^7$$

3.5 Hjälp till Maple i Maple

Maple innehåller en lättillgänglig manual. Om du skriver `?library` får du en lista över de standarfunktioner som Maple arbetar med (dvs en lista som bland annat innehåller namnen på de funktioner som räknas upp i tabellerna på näst-följande sidor). Vill du ha information om en speciell Maplefunktion, t e x rit-kommandot `plot` skriver du bara `?plot` så får du en kortare beskrivning av plotfunktionen inklusive en del exempel.

3.6 Att använda de resultat som Maple producerat

Vill man använda resultatet i senaste beräkningen anger man `%` som argument. Näst senaste resultat kan nås genom att ange `%%` o s v.

3.7 Egenheter hos Maple

Om man ber Maple rita en funktion på ett intervall och det visar sig att den ej är definierad på hela detta intervall, ritas funktionen i den del av intervallet där den är definierad. En funktion som ser ut att vara noll på ett delintervall kan således vara odefinierad där, det får användaren reda ut själv.

Vill man lösa en ekvation, t ex $f(x) = 0$ är det lämpligt att tilldela lösningen ett namn, `s:=solve(f(x) = 0, x);` Om Maple hittar fler lösningar till ekvationen kommer dessa att lagras som en sträng, och de lösningar Maple skriver upp kommer i tur och ordning att få namnen `s[1]` , `s[2]` o s v.

Vid lösning av polynomekvationer av högst grad 4 ger Maple alla lösningar, men vid lösning av andra ekvationer är det ganska vanligt att Maple endast anger en del lösningar. Ibland svarar Maple “blankt”, d v s man erhåller inget svar alls. Detta betyder att Maple inte klarar av att lösa ekvationen. Om detta sedan beror på att lösning saknas, eller på brister i Maple, för användaren själv hålla reda på. Tyvärr producerar Maple ibland felaktiga svar, så ett visst mått av försiktighet måste iakttagas.

4 Några kommandon i Maple

En fullständig förteckning över alla maplefunktioner och kommandon finner man i standardverken [4] och [3].

Kommando	**Betydelse**
`assume(x,x>0)`	talar om för Maple att x är positivt
`collect(f,x)`	samlar koefficienter framför likadana gradtal
`combine(f)`	samlar ihop termer, gör omvändningen till `expand(f)`
`combine(f,n)`	där `n` är något av uttrycken `exp`, `ln`, `power` eller `trig`, behövs ibland för att hjälpa `combine` att hitta räknelagar
`convert(f,form)`	omvandlar f till alternativ form (se `?convert`)
`eval(f,x=a)`	ersätter x med a i variabeln f
`expand(f)`	utvecklar trigonometriska uttryck, produkter av polynom m m
`factor(f)`	faktoriserar polynom och rationella funktioner i faktorer med rationella koefficienter
`normal(f)`	förenklar rationella uttryck och faktoriserar nämnaren
`simplify(f)`	”förenklar” uttrycket f med hjälp av vanliga räknelagar
`sort(p)`	sorterar polynom i avtagande gradtal
`coeff(p, x, n)`	anger koefficienten för x^n i polynomet p (som måste vara sorterat)
`coeffs(p, x)`	anger alla koefficienter i polynomet p (efter stigande gradtal)
`lhs(f=g)`	vänsterled i ekvationen $f = g$
`rhs(f=g)`	högerled i ekvationen $f = g$
`quo(a,b,x)`	kvot då a delas med b (a och b polynom i x)
`rem(a,b,x)`	rest då a delas med b (a och b polynom i x)
`denom(f)`	nämnaren (denominator) i funktionen f
`numer(f)`	täljaren (numerator) i funktionen f
`ifactor(n)`	primtalsfaktoriserar heltalet n
`convert(z,polar)`	skriver komplexa talet z på polär form

Kommando	**Betydelse**
`evalc(a)`	Uttrycker det komplexa talet a på formen $x + iy$
`evalf(a)`	beräknar ett närmevärde till talet a
`evalf(a,n)`	beräknar ett närmevärde till talet a med n siffror
`fsolve(f=g,x)`	löser ekvationen $f = g$ med avseende på x numeriskt. Maple finner normalt närmevärde till en reell rot om inte f och g är polynom
`solve(f=g,x)`	löser ekvationen $f = g$ med avseende på variabeln x. Maple finner samtliga lösningar till polynomekvationer av högst grad 4 (även komplexa lösningar), däremot är det inte säkert att programmet finner alla lösningar då andra ekvationer betraktas. Det händer t o m att falska lösningar presenteras
`plot(f,x=a..b)`	ritar kurvan $y = f(x)$, $a \leq x \leq b$
`plot(f,x=a..b,y=c..d)`	som ovan, men $y = f(x)$ ritas bara mellan c och d
`plot({f,g},x=a..b)`	ritar $y = f(x)$ och $y = g(x)$ i samma figur
`diff(f,x)`	deriverar f med avseende på x
`dsolve(ekv,y(x))`	löser differentialekvationer (se `?dsolve`)
`int(f,x)`	beräknar en primitiv funktion till f m a p x
`int(f,x=a..b)`	beräknar $\int_a^b f(x)\,dx$
`sum(f,k=m..n)`	beräknar $\sum_{k=m}^{n} f(k)$
`product(f,k=m..n)`	beräknar $\prod_{k=m}^{n} f(k)$
`limit(f,x=a)`	beräknar $\lim_{x\to a} f(x)$ (`a` kan även vara `infinity` eller `-infinity`)
`taylor(f,x=a)`	taylorutveckling av f runt punkten a, resttermen av grad 6
`taylor(f,x=a,n)`	taylorutveckling av f runt punkten a, resttermen av grad n

5 Grundläggande formelbehandling

Här är några fler exempel med kommentarer.

5.1 Rationella funktioner

En rationell funktion

```
> f := (x^4+4*x^3+6*x^2+4*x+1)/(x^3+6*x^2+11*x+6);
```

$$f := \frac{x^4 + 4\,x^3 + 6\,x^2 + 4\,x + 1}{x^3 + 6\,x^2 + 11\,x + 6}$$

förenklas

```
> simplify(f);
```

$$\frac{x^3 + 3\,x^2 + 3\,x + 1}{x^2 + 5\,x + 6}$$

genom att gemensamma faktorer förkortas. Nästa kommando faktoriserar täljare och nämnare.

```
> factor(f);
```

$$\frac{(x+1)^3}{(x+3)\,(x+2)}$$

Täljaren i senaste uttrycket.

```
> numer(%);
```

$$(x+1)^3$$

Nämnaren i näst senaste uttrycket.

```
> denom(%%);
```

$$(x+3)\,(x+2)$$

Slutligen partialbråksuppdelning.

```
> convert(f, parfrac, x);
```

$$x - 2 + 8\,\frac{1}{x+3} - \frac{1}{x+2}$$

En annan rationell funktion.

```
> g := 1/x+1/(x+1)+1/(x^2+4*x+3);
```

$$g := \frac{1}{x} + \frac{1}{x+1} + \frac{1}{x^2 + 4\,x + 3}$$

Kommandot `normal` skriver g som ett bråk och förenklar och faktoriserar nämnaren.

```
> normal(g);
```

$$\frac{2\,x^2 + 8\,x + 3}{x\,(x^2 + 4\,x + 3)}$$

Beräkna g-s värde i $x = 1$

```
> eval(g, x=1);
```

$$\frac{13}{8}$$

Även numeriskt

```
> evalf(%);
```

$$1.625000000$$

5.2 Polynom

Ett polynom

```
> h := (x^2+4*x+7)*(x^2+2*x+1)+x^3+2*x+9;
```

$$h := (x^2 + 4\,x + 7)\,(x^2 + 2\,x + 1) + x^3 + 2\,x + 9$$

utvecklat som summa

```
> expand(h);
```

$$x^4 + 7\,x^3 + 16\,x^2 + 20\,x + 16$$

och faktoriserat i reella faktorer

```
> factor(h);
```

$$(x + 4)\,(x + 2)\,(x^2 + x + 2)$$

Ett polynom till

```
> j := x^3+3*x;
```

$$j := x^3 + 3\,x$$

Funktionen `rem` ger resten vid divisionen h/j

```
> rem(h, j, x, 'q');
```

$$16 + 13\,x^2 - x$$

Kvoten lagras i variabeln q

```
> q;
```

$$x + 7$$

Kvoten kan också beräknas direkt med funktionen `quo(h, j, x)` Koefficienterna behöver inte vara numeriska

```
> h1 := (a+b)*(x^2+1)+(a+c)*(x^2+2*x)+b*(x+1);
```

$$h1 := (a + b)\,(x^2 + 1) + (a + c)\,(x^2 + 2\,x) + b\,(x + 1)$$

Utveckla $h1$

```
> expand(h1);
```

$$2\,a\,x^2 + a + b\,x^2 + 2\,b + 2\,a\,x + c\,x^2 + 2\,c\,x + b\,x$$

Sortera föregående uttryck i fallande gradtal i x

```
> sort(%, x);
```

$$2\,a\,x^2 + c\,x^2 + b\,x^2 + b\,x + 2\,a\,x + 2\,c\,x + a + 2\,b$$

Samla koefficienterna framför lika gradtal.

```
> collect(%, x);
```

$$(2\,a + c + b)\,x^2 + (2\,a + 2\,c + b)\,x + a + 2\,b$$

5.3 Andra uttryck

Även ett trigonometriskt uttryck

```
> f1 := sin(x+2*y);
```

$$\mathit{f1} := \sin(x + 2\,y)$$

kan utvecklas med `expand`

```
> expand(f1);
```

$$2\sin(x)\cos(y)^2 - \sin(x) + 2\cos(x)\sin(y)\cos(y)$$

För att göra tvärtom använder vi `combine`.

```
> g1 := cos(x)^4-sin(x)^4;
```

$$\mathit{g1} := \cos(x)^4 - \sin(x)^4$$

```
> combine(g1);
```

$$\cos(2\,x)$$

Maple kan ibland hjälpas på traven med `combine(g1, trig)`

5.4 Ekvationer

```
> s := solve(2*x^4+15*x^3+35*x^2+19*x-15 = 0, x);
```

$$s := \frac{-5}{2},\ -3,\ -1 + \sqrt{2},\ -1 - \sqrt{2}$$

Variabeln s tilldelas lösningarna till $2x^4 + 15x^3 + 35x^2 + 19x - 15 = 0$

```
> s[3];
```

$$-1 + \sqrt{2}$$

Väljer ut den tredje av lösningarna ovan

```
> evalf(s[3]);
```

$$.414213562$$

Närmevärde till s_3.
Mer komplicerade ekvationer som

```
> solve(exp(x)=7+ln(x),x);
```

$$e^{\mathrm{RootOf}(\_Z - e^{(e^{\_Z})} + 7)}$$

klarar Maple inte av att lösa exakt. Men det gör oftast ingen annan heller. Vi kan däremot enkelt bestämma närmevärden till eventuella lösningar. Först ritar vi funktionerna e^x och $7 + \ln x$ för $0 \leq x \leq 3$.

```
> plot({exp(x),7+ln(x)}, x=0..3);
```



Figur 1: Funktionerna $y = e^x$ och $y = 7 + \ln x$

Det finns antagligen en lösning nära $x = 0$ också. Vi ritar funktionerna i intervallet $0 \leq x \leq 0.01$.

```
> plot({exp(x),7+ln(x)}, x=0..0.01);
```



Samma funktioner som i figur 1 fast nära $x = 0$

Funktionen `fsolve` hittar ett närmevärde till den minsta roten.

```
> fsolve(exp(x) = 7+ln(x), x);
```

$$.002484927033$$

Den största får vi genom att begränsa intervallet till $0 \leq x \leq 3$.

```
> fsolve(exp(x) = 7+ln(x), x=1..3);
```

$$2.043097396$$

6 Lineär algebra

6.1 Något om datastrukturer i Maple

Hanteringen av vektorer och matriser i Maple är tyvärr inte så intuitiv som man kunde önska. En vektor[1] $x = (x_1,\ x_2,\ x_3)$ kan uppfattas som en ordnad följd av tal: koordinater. Maple har flera datastrukturer för ordnade följder. De som främst intresserar oss nu är

1. Sekvenser: $x_1,\ x_2,\ x_3$
2. Listor (`list`): $[x_1,\ x_2,\ x_3]$ och
3. Fält (`array`): $\mathrm{array}\,(1..3,\ [x_1,\ x_2,\ x_3])$

Låt oss exemplifiera. Vi börjar med en sekvens S

```
> S := 1, 2, a, sin(t);
```

$$S := 1,\, 2,\, a,\, \sin(t)$$

De enskilda elementen S_j, där $j = 1, \ldots, 4$, får vi som

```
> S[4];
```

$$\sin(t)$$

En lista L är en sekvens med hakparenteser omkring

```
> L := [S];
```

$$L := [1,\, 2,\, a,\, \sin(t)]$$

De enskilda elementen kommer vi åt på samma sätt som för sekvenser

```
> L[3];
```

$$a$$

Ett fält är en lista där man i någon mån själv kan påverka indiceringen

```
> A := array(1..4, L);
```

$$A := [1,\, 2,\, a,\, \sin(t)]$$

Även här ger hakparenteserna de enskilda elementen

```
> A[3];
```

$$a$$

Sekvenser och listor evalueras direkt vid anrop

```
> S;
```

$$1,\, 2,\, a,\, \sin(t)$$

```
> L;
```

[1] Att det också finns vektorer med oändlig dimension bortser vi helt från i detta sammanhang.

$$[1, 2, a, \sin(t)]$$

Detta gäller emellertid inte fält

```
> A;
```

$$A$$

Funktionerna `print` och `evalm` (evaluate matrix) ger värdet av A

```
> print(A);
```

$$[1, 2, a, \sin(t)]$$

```
> evalm(A);
```

$$[1, 2, a, \sin(t)]$$

Listor och fält, men inte sekvenser, kan användas som parametrar till vissa funktioner. Funktionen `map` är ofta användbar.

```
> map(exp, L);
```

$$[e, e^2, e^a, e^{\sin}(t)]$$

```
> f := x -> x^2: map(f, A);
```

$$[1, 4, a^2, \sin(t)^2]$$

För att krångla till det ytterligare finns dessutom mängder (`set`) $\{x_1,\ x_2,\ x_3\}$ och tabeller (`table`).

6.2 Vektorer

I Maple representeras vektorer och matriser som fält. Det finns dessutom ett särskilt paket, `linalg`, med kommandon för att räkna med vektorer och matriser. Paketet laddas med funktionen `with` och ger en lång lista med nya funktioner och kommandon.

```
> with(linalg):
```

```
Warning, new definition for norm

Warning, new definition for trace
```

6.2.1 Exempel

Två vektorer

```
> x := vector([1,2,3,a]);
```

$$x := [1, 2, 3, a]$$

```
> y := vector([b,-1,1,0]);
```

$$y := [b, -1, 1, 0]$$

Lineärkombinationer kan bildas på vanligt sätt

```
> 5*x-y/2;
```

$$5\,x - \frac{1}{2}\,y$$

men vill vi utföra beräkningen och lagra resultatet som en vektor z måste vi använda `evalm`

```
> z := evalm(%);
```

$$z := \left[5 - \frac{1}{2}\,b,\ \frac{21}{2},\ \frac{29}{2},\ 5\,a\right]$$

6.3 Matriser

Matriser är fält som är indicerade med talpar. Konstruktionen av en 3×3-matris, $A = \begin{bmatrix} 1 & 0 & -1 \\ 0 & 2 & 0 \\ -2 & 1 & 0 \end{bmatrix}$, går till på så sätt att vi med `array`-funktionen indicerar en lista av listor L

```
> L := [[1,0,-1],[0,2,0],[-2,1,0]];
```

$$L := [[1,\ 0,\ -1],\ [0,\ 2,\ 0],\ [-2,\ 1,\ 0]]$$

med talpar i, j som båda löper från 1 till 3 och sedan lagrar resultatet som ett fält A.

```
> A := array(1..3,1..3,L);
```

$$A := \begin{bmatrix} 1 & 0 & -1 \\ 0 & 2 & 0 \\ -2 & 1 & 0 \end{bmatrix}$$

Elementen $A_{i,j}$ får vi som

```
> A[3,1];
```

$$-2$$

Det är dock enklare att använda de särskilda funktioner som finns i `linalg`

```
> B := matrix(3,3,[1,2,3,1,2,0,-1,1,0]);
```

$$B := \begin{bmatrix} 1 & 2 & 3 \\ 1 & 2 & 0 \\ -1 & 1 & 0 \end{bmatrix}$$

```
> E := diag(1,1,1);
```

$$E := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

Matrismultiplikation görs med operatorn $\&*$

```
> A&*B;
```

$$A \mathbin{\&*} B$$

men för att få resultatet behöver vi `evalm`

```
> evalm(%);
```

$$\begin{bmatrix} 2 & 1 & 3 \\ 2 & 4 & 0 \\ -1 & -2 & -6 \end{bmatrix}$$

Matriser och vektorer $x = (x_1, x_2, x_3)$ av rätt dimension kan också multipliceras

```
> x := vector(3); y := evalm(A&*x);
```

$$x := \mathrm{array}(1..3, [])$$

$$y := [x_1 - x_3,\ 2\,x_2,\ -2\,x_1 + x_2]$$

Determinanter beräknas med funktionen `det`

```
> det(s*E-A);
```

$$s^3 - 3\,s^2 + 4$$

och inversmatriser med `inverse`

```
> inverse(s*E-A);
```

$$\begin{bmatrix} \dfrac{s}{s^2 - s - 2} & -\dfrac{1}{s^3 - 3\,s^2 + 4} & -\dfrac{1}{s^2 - s - 2} \\ 0 & \dfrac{1}{s-2} & 0 \\ -2\,\dfrac{1}{s^2 - s - 2} & \dfrac{s-1}{s^3 - 3\,s^2 + 4} & \dfrac{s-1}{s^2 - s - 2} \end{bmatrix}$$

För att sätta ihop två eller flera matriser A och B används funktionerna `augment(A,B)` eller `concat(A,B)` som ger matrisen

$$\begin{bmatrix} A & B \end{bmatrix}$$

eller `stackmatrix(A,B)` som ger

$$\begin{bmatrix} A \\ B \end{bmatrix}$$

allt givetvis under förutsättning att A och B har lämpliga format. Strukturen `vector` kan Maple tolka både som kolonn och rad vilket många gånger kan vara bekvämt. Vi illustrerar med ett par exempel med anknytning till reglerteknik

```
> A := matrix(3,3,[1,2,3,4,5,6,7,8,9]); B := vector([1,5,9]);
```

$$A := \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

$$B := [1,\ 5,\ 9]$$

```
> augment(B, A&*B, A^2&*B);
```

$$\begin{bmatrix} 1 & 38 & 588 \\ 5 & 83 & 1335 \\ 9 & 128 & 2082 \end{bmatrix}$$

```
> stackmatrix(B, A&*B, A^2&*B);
```

$$\begin{bmatrix} 1 & 5 & 9 \\ 38 & 83 & 128 \\ 588 & 1335 & 2082 \end{bmatrix}$$

Funktionerna `extend` och `copyinto` kan ibland också vara användbara. För närmare detaljer hänvisas till onlinemanualen i Maple.

7 Differentialekvationer

Maple är ett mycket användbart vektyg för att studera och lösa differentialekvationer. I enkla fall kan man till och med få exakta lösningar i form av elementära funktioner men även när detta inte går, har man stor nytta av att grafiskt visualisera vad som händer.

7.1 Riktningsfält

En differentialekvation av första ordningen

$$y' = f(t, y)$$

kan uppfattas som ett *riktningsfält* där det i varje punkt (t, y) finns en riktning given av funktionen $f(t, y)$. Detta åskådligörs genom att i punkten rita en pil med lutningen $f(t, y)$. För att rita dessa pilar använder vi maplepaketet `DEtools`

```
> with(DEtools):
```

7.1.1 Exempel

Vi illustrerar med ett enkelt exempel

$$y' = 4t\sqrt{y} \qquad (1)$$

```
> eq := diff(y(t),t)=4*t*sqrt(y(t));
```

$$eq := \frac{\partial}{\partial t}\,\mathrm{y}(t) = 4\,t\,\sqrt{\mathrm{y}(t)}$$

Riktningsfältet ritas sedan med DEplot

```
> DEplot(eq, y(t),  t=-1..1, y=-1..4, arrows=MEDIUM);
```



Riktningsfält till ekvation (1).

En lösning som uppfyller villkoret $y\,(1/2) = 2$ får vi med kommandot

```
> DEplot(eq, y(t), t=-1..1, [[y(1/2)=2]], y=-1..4, arrows=MEDIUM,
> color=GREEN, linecolor=RED);
```



Lösning genom punkten $(1/2, 2)$.

7.1.2 Ännu ett exempel

Som ett mer komplicerat exempel betraktar vi ekvationen

$$y' = 3t^2\left(1 - y^2\right) \tag{2}$$

Riktningsfältet blir

```
> eq := diff(y(t),t)=3*t^2*(1-y(t)^2);
```

$$eq := \frac{\partial}{\partial t}\,\mathrm{y}(t) = 3\,t^2\,(1 - \mathrm{y}(t)^2)$$

```
> DEplot(eq, y(t),  t=-2..2, y=-3..4, arrows=MEDIUM);
```



Riktningsfält till ekvation (2).

Ritar vi dessutom in lösningar genom olika punkter får vi

```
> inits := {[0,1],[0,0.5],[0,1.5],[0,-1.5],[2,2]};
```

$$inits := \{[0, 1], [0, .5], [0, 1.5], [0, -1.5], [2, 2]\}$$

```
> DEplot(eq, y(t),  t=-2..2, y=-3..4, inits, stepsize=0.1,
> arrows=MEDIUM, color=GREEN, linecolor=RED);
```



7.2 Exakta lösningar

De enkla ekvationerna (1) och (2) ovan klarar Maple också att lösa exakt. Då används funktionen `dsolve` och vi får

```
> eq1 := diff(y(t),t)=4*t*sqrt(y(t));
```

$$eq1 := \frac{\partial}{\partial t}\,\mathrm{y}(t) = 4\,t\,\sqrt{\mathrm{y}(t)}$$

```
> dsolve(eq1, y(t));
```

$$\sqrt{y(t)} - t^2 - \_C1 = 0$$

Maple löser inte ut funktionen $y(t)$ eftersom det är oklart vilken rot som avses. Tvingar vi fram en *explicit* lösning får vi

```
> dsolve(eq1, y(t), explicit);
```

$$y(t) = \mathrm{RootOf}(\sqrt{\_Z} - t^2 - \_C1)$$

vilket inte är så mycket bättre.
Ekvationen i nästa exempel ger däremot en explicit lösning

```
> eq2 := diff(y(t),t)=3*t^2*(1-y(t)^2);
```

$$eq2 := \frac{\partial}{\partial t}\, y(t) = 3\, t^2\, (1 - y(t)^2)$$

```
> dsolve(eq2, y(t));
```

$$y(t) = \frac{e^{(2\, t^3)}\ \_C1 + 1}{-1 + e^{(2\, t^3)}\ \_C1}$$

Vi kan emellertid av olika skäl vilja ha lösningen i *implicit* form och kan då använda kommandot

```
> dsolve(eq2, y(t), implicit);
```

$$\frac{1}{3}\, t^3 + \frac{1}{6} \ln(y(t) - 1) - \frac{1}{6} \ln(y(t) + 1) + \_C1 = 0$$

7.3 Nivåkurvor

Ett skäl att föredra lösningen i implicit form är att man då enkelt kan åskådliggöra den med nivåkurvor. För ekvation (2) fann vi att lösningarna ligger på nivåkurvorna till funktionen

$$F(t, y) = 2t^3 + \ln(y - 1) - \ln(y + 1) = 2t^3 + \ln\left(\frac{y-1}{y+1}\right)$$

8 Laplacetransform

Ett annat paket i Maple är `inttrans` (integral transforms) som bland annat innehåller enkelsidig laplacetransform

$$F(s) = \int_0^{\infty} e^{-st} f(t)\, dt$$

```
> with(linalg): with(inttrans);
```

```
Warning, new definition for norm

Warning, new definition for trace

Warning, new definition for hilbert
```

$$[addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable]$$

För att beräkna laplacetransformen till funktionen

```
> f := exp(-2*t)*sin(omega*t);
```

$$f := e^{(-2\,t)}\sin(\omega\,t)$$

skriver vi

```
> F := laplace(f,t,s);
```

$$F := \frac{\omega}{(s+2)^2+\omega^2}$$

Vill vi sedan ha den inversa transformen av $sF\,(s)$ använder vi `invlaplace`

```
> invlaplace(s*F,s,t);
```

$$\omega\,(e^{(-2\,t)}\cos(\omega\,t) - 2\,\frac{e^{(-2\,t)}\sin(\omega\,t)}{\omega})$$

8.1 Begynnelsevärdesproblem

Det är främst för att lösa lineära differentialekvationer med konstanta koefficienter och givna begynnelsedata som vi skall använda laplacetransformen. Följande exempel får illustrera den hanteringen. Se även [1, Kap. 9]. En utförligare behandling av differentialekvationer i Maple finner man i [5] och [6].

8.1.1 Exempel

Lös begynnelsevärdesproblemet

$$\begin{cases} y'' + 2y' + 5y = u\,(t) \\ y\,(0) = 1,\ y'\,(0) = 0 \end{cases}$$

där högerledet $u\,(t)$ definieras som

$$u\,(t) = \begin{cases} \sin t, & 0 \le t \le 2\pi \\ 0, & \text{f.ö.} \end{cases}$$

Mata in ekvationen

```
> eqn := diff(y(t),t$2)+2*diff(y(t),t)+5*y(t) = u(t);
```

$$eqn := (\frac{\partial^2}{\partial t^2}\,\mathrm{y}(t)) + 2\,(\frac{\partial}{\partial t}\,\mathrm{y}(t)) + 5\,\mathrm{y}(t) = \mathrm{u}(t)$$

och högerledet som beskrives med förskjutna stegfunktioner

```
> u := t -> (1-Heaviside(t-2*Pi))*sin(t);
```

$$u := t \to (1 - \mathrm{Heaviside}(t - 2\,\pi))\sin(t)$$

Kontrollera att resultatet blev det förväntade

```
> plot(u(t),t=0..10);
```



Funktionen $u(t)$

Laplacetransformera

```
> Leqn := laplace(eqn,t,s);
```

$$Leqn := s\,(s\,\mathrm{laplace}(\mathrm{y}(t),\,t,\,s) - \mathrm{y}(0)) - \mathrm{D}(y)(0) + 2\,s\,\mathrm{laplace}(\mathrm{y}(t),\,t,\,s) - 2\,\mathrm{y}(0) + 5\,\mathrm{laplace}(\mathrm{y}(t),\,t,\,s) = \frac{1}{s^2+1} - \frac{e^{(-2\,s\,\pi)}}{s^2+1}$$

Lös ut laplacetransformen av lösningen till differentialekvationen och kalla den Y

```
> Y := solve(Leqn,laplace(y(t),t,s));
```

$$Y := \frac{\mathrm{y}(0)\,s^3 + s\,\mathrm{y}(0) + \mathrm{D}(y)(0)\,s^2 + \mathrm{D}(y)(0) + 2\,\mathrm{y}(0)\,s^2 + 2\,\mathrm{y}(0) + 1 - e^{(-2\,s\,\pi)}}{s^4 + 6\,s^2 + 2\,s^3 + 2\,s + 5}$$

Substituera begynnelsedata

```
> Y := eval(Y, [y(0)=1, D(y)(0)=0]);
```

$$Y := \frac{s^3 + s + 3 + 2\,s^2 - e^{(-2\,s\,\pi)}}{s^4 + 6\,s^2 + 2\,s^3 + 2\,s + 5}$$

Inverstransformera

```
> invlaplace(Y,s,t);
```

$$-\frac{1}{10}\cos(t) + \frac{1}{5}\sin(t) + \frac{11}{10}\,e^{(-t)}\cos(2\,t) + \frac{9}{20}\,e^{(-t)}\sin(2\,t) - \frac{1}{10}\,\%1\,e^{(-t+2\,\pi)}\cos(2\,t) + \frac{1}{20}\,\%1\,e^{(-t+2\,\pi)}\sin(2\,t) + \frac{1}{10}\,\%1\cos(t) - \frac{1}{5}\sin(t)\,\%1$$

$$\%1 := \mathrm{Heaviside}(t - 2\,\pi)$$

Definiera funktionen $y(t)$

```
> y := unapply(%,t);
```

$$y := t \to -\frac{1}{10}\cos(t) + \frac{1}{5}\sin(t) + \frac{11}{10}\,e^{(-t)}\cos(2\,t) + \frac{9}{20}\,e^{(-t)}\sin(2\,t) - \frac{1}{10}\,\mathrm{Heaviside}(t - 2\,\pi)\,e^{(-t+2\,\pi)}\cos(2\,t) + \frac{1}{20}\,\mathrm{Heaviside}(t - 2\,\pi)\,e^{(-t+2\,\pi)}\sin(2\,t) + \frac{1}{10}\,\mathrm{Heaviside}(t - 2\,\pi)\cos(t) - \frac{1}{5}\sin(t)\,\mathrm{Heaviside}(t - 2\,\pi)$$

Jämför slutligen $u(t)$ och $y(t)$

```
> plot({u(t),y(t)},t=0..10);
```



Funktionerna $u(t)$ och $y(t)$

8.2 Vektorer och matriser

För att laplacetransformera vektorer och matriser kan vi använda `map` funktionen. Med matrisen $R = (sE - A)^{-1}$ ovan får vi som exempel

```
> map(invlaplace, R, s, t);
```

$$\begin{bmatrix} \frac{1}{3}e^{(-t)} + \frac{2}{3}e^{(2t)} & -\frac{1}{9}e^{(-t)} - \frac{1}{3}t\,e^{(2t)} + \frac{1}{9}e^{(2t)} & -\frac{1}{3}e^{(2t)} + \frac{1}{3}e^{(-t)} \\ 0 & e^{(2t)} & 0 \\ -\frac{2}{3}e^{(2t)} + \frac{2}{3}e^{(-t)} & -\frac{2}{9}e^{(-t)} + \frac{1}{3}t\,e^{(2t)} + \frac{2}{9}e^{(2t)} & \frac{2}{3}e^{(-t)} + \frac{1}{3}e^{(2t)} \end{bmatrix}$$

Matrisen ovan är lösningen till begynnelsevärdesproblemet

$$X'(t) = AX(t)\,,\ X(0) = E$$

för matris-differentialekvationen $X' = AX$ ty laplacetransformering av ekvationen ger

$$s\widetilde{X}(s) = A\widetilde{X}(s) + X(0) = A\widetilde{X}(s) + E$$

och om vi bryter ut $\widetilde{X}(s)$ och löser får vi

$$(sE - A)\widetilde{X}(s) = E \Leftrightarrow \widetilde{X}(s) = (sE - A)^{-1} = R$$

Matrisfunktionen $X(t)$ kallas exponentialmatrisen till A och betecknas $X(t) = e^{tA}$. Den kan också erhållas direkt med funktionen `exponential` som ingår i `linalg`-paketet

```
> exponential(t*A);
```

$$\begin{bmatrix} \frac{2}{3}e^{(2t)} + \frac{1}{3}e^{(-t)} & -\frac{1}{9}e^{(-t)} + \frac{1}{9}e^{(2t)} - \frac{1}{3}t\,e^{(2t)} & \frac{1}{3}e^{(-t)} - \frac{1}{3}e^{(2t)} \\ 0 & e^{(2t)} & 0 \\ \frac{2}{3}e^{(-t)} - \frac{2}{3}e^{(2t)} & \frac{1}{3}t\,e^{(2t)} - \frac{2}{9}e^{(-t)} + \frac{2}{9}e^{(2t)} & \frac{1}{3}e^{(2t)} + \frac{2}{3}e^{(-t)} \end{bmatrix}$$

Referenser

[1] Göran Andersson: *Tillämpad matematik med Maple,* Studentlitteratur 1995.

[2] Bruce W. Char m.fl.: *Maple V: First Leaves*, Springer-Verlag 1992.

[3] Bruce W. Char m.fl.: *Maple V: Programming guide,* Springer-Verlag 1998.

[4] Bruce W. Char m.fl.: *Maple V: Library Reference Manual,* Springer-Verlag 1992.

[5] Jon H. Davis: *Differential Equations with Maple,* Birkhäuser 2001.

[6] Darren Redfern, Edgar Chandler: *Maple O.D.E. Lab Book,* Springer-Verlag 1996.